query: SELECT ?item WHERE {
  BIND(wd:Q34 AS ?country)
  {
  VALUES ?type {
  wd:Q68295960 # Swedish government agency (249) ✔
  wd:Q107407151 # Swedish government agency (5) ✔
  wd:Q127448 # municipality of Sweden (290) ✔
  wd:Q1754161 # regional council in Sweden (20) ✔
  wd:Q10397683 # AP-fund (6) ✔
  wd:Q59603261 # state-ownded enterprise (3) ✔
  wd:Q10330441 # state level departments (11) ✔
  wd:Q341627 # court of appeal (6) ✔
  wd:Q2065704 # tingsrätt (48) ✔
  wd:Q190752 # supreme court (2) ✔
  wd:Q1289455 # administrative court of appeal (4) ✔
  wd:Q18292311 # migration court (4) ✔
  wd:Q10530889 # regional rent tribunal (8) ✔
  }
  ?item wdt:P31 ?type; wdt:P17 ?country .
  }
  UNION
  {
  VALUES ?type {
  wd:Q108059166 # delegations (7) ✔
  wd:Q108058047 # general consulates (9) ✔
  wd:Q3917681 # embassies (93) ✔
  }
  ?item wdt:P31 ?type; wdt:P137 ?country .
  }
  UNION
  {
  VALUES ?item {
  wd:Q10475844 # domstolsverket
  }
  }
  ?item wdt:P31 ?type
  MINUS { ?item wdt:P576 [] }
  MINUS { ?item wdt:P1366 [] }
  MINUS { ?item wdt:P3999 [] } 
  }